

Title	PASOPIAのTSS端末プログラム
Author(s)	柳瀬, 章
Citation	大阪大学大型計算機センターニュース. 1983, 49, p. 77-86
Version Type	VoR
URL	https://hdl.handle.net/11094/65566
rights	
Note	

Osaka University Knowledge Archive : OUKA

<https://ir.library.osaka-u.ac.jp/>

Osaka University

PASOPIAのTSS端末プログラム

大阪府立大学 総合科学部 柳 瀬 章

1. はじめに

端末専用機とはちがって、パーソナルコンピューターを、TSS端末として使用すると、利用者の好みに合わせて端末としての機能を持たせることができる。しかしこの反面パーソナルコンピューターのプログラムを作らなければならない面倒さがある。今回パソピアを端末として利用するBASICプログラムを作製したので紹介する。

市販されているパーソナルコンピューターは互いに共通の機能を持っているが、それぞれに特徴を持っており、TSS端末として利用する時にも、その機種、システムに合った工夫をする必要がある。パソピアはT-BASICとOA-BASICの二種類のBASICを持っているが、ここでは、T-BASICを用いている。このシステムに特徴的な点に重点をおいて、以下の記述をする。

2. 機器構成とプログラムの機能

機器構成を表1に示す。PA7251のプリンターは9インチの中の中に圧縮文字を使って、136字/行を書くことができる。これはバッチ出力をTSSからのぞく時等に便利であり、このプログラムでも、この機能を活かせるようにした。パソピアはドット情報を画面に出すことができるので、これを使ってプログラム走行時にカーソルを点滅させることができる。カーソルのパターンを変えて、プリンターのON/OFFが画面上でわかるようにしておくことにした。

端末の入出力を、プリンターにその時点で出すだけでなく、プリンターをOFFにしてホストと交信して画面上に出てしまった文字を、プリンターにハードコピーがとれると便利である。T-BASICの機能を用いてこのプログラムでは、これを可能にしている。BASICプログラムが、“LINE INPUT”文に達するとパソコンのスクリーンエディットができるようになる。ACOSのTSSで“EDITOR”SYSTEMを使って、ファイルを修正する時のためにこの機能を使えるようにしてある。

パソピアには8個のファンクションキーがある。各キーには最大15字迄の文字列が定義できる他、割込みキーとして使用できる。このプログラムでは **PF1** をTSSの“Break”信号発生用に **PF2** をこのプログラムの修了作業に **PF3** をプリンターのON/OFF制御にそれぞれ用いている。 **PF4** はフロッピーディスクからテキストを送り出す時に **PF5** はフロッピーディスクにテキストを書き込む時に使用する。 **PF7** にはもともと“LIST”の

文字が **PF8** には“ RUN (復掃) ”の文字が入っているので、これはそのまま使用する。

PF6 は使っていない。TSS 端末が入力待ちになると、プログラムは空まわりの状態になる。この時キーボードから特殊文字を入力させることで、一種の割り込み機能を持たせることができる。このプログラムでは“ (CTR) Q ”を画面のハードコピーを取るために、“ (CTR) A ”をログオン手続きの発生に使っている。“ (CTR) C ”はT-BASICが走行中のプログラムを中断するのに使用しているのでTSSの端末としてのDisconnect信号としては使えない。このプログラムでは“ (CTR) E ”をこの代用に使えるようにしてある。

3. プログラムの構成と使用法

プログラム全体を図1に示す。大きくわけて ①初期化 ②送信ルーチン ③受信ルーチン ④割り込み処理ルーチン ⑤フロッピーディスクの入出力ルーチンからなっている。以下プログラムの流れにそって順に説明する。

① 初期化

- 行 40,50 カーソル用のドットデータをあたえる。
- 行 60 プリンターをOFFにする。
- 行 70 ハードコピーのサブルーチンのある番地 (行 590 以下参照)
- 行 80 Break 文字 (行 440 以下参照)
- 行 90,100 ファンクションキーに割り込み機能をあたえる。
- 行 110 エラー処理のルーチンを設定する (行 500 以下参照)
- 行 120 画面の印字巾を80字に、画面を消去し、カーソルのために画面をグラフィックモードにする。送受信の口であるRS 232Cを動作状態にする。-3 はバソピア本体のRS 232Cを指し、300 は送受信速度を300ボーに、Eは偶数パリティ、71 は一文字7bit、ストップビットを1bitにすることを意味している。

② 送信ルーチン

プログラムがここまで進むとカーソルがチラチラ見えはじめ、端末として同作状態に入ったことがわかる。

- 行 140 KEY入力から一文字とり出す。
- 行 150 入力が空ならば、カーソルを一回点滅させて受信ルーチンへ。
- 行 160,170 入力が“ ← ”、“ (CTR) H”、又は“ (DEL) ”ならばBackspaceルーチン410へ。
- 行 180 入力が“ (CTR) E ”ならばDisconnectルーチンへ。
- 行 190 “ → ”入力を無効にする。
- 行 200 “ ↑ ”入力ならば“ LINE INPUT ”入力ができるように 680 行へ。

- 行 210 “ ↓ ”入力を無効にする。
- 行 220 “ (CTR) A ”入力ならば、ログオン文字列をホスに送るルーチン 580 行へ。
- 行 230 “ (CTR) Q ”入力ならば、ハードコピールーチン 600 行へ。
- 行 240 入力が 150 ~ 230 行のチェックにかからなければ、ホストに一字送り出す。パルピアでは送信文字が受信バッファにも入ってしまうので、今出した文字を空読みする操作をつづけて行う。
- 行 250 入力文字が“ 復帰 ”の場合はそのまま画面上に出すと、カーソルルーチンのために行の最初の文字が消されてしまう。このことをふせぐため“ 改行 ”を前につけ加える。
- 行 260 画面上に入力文字を出す。
- 行 270 プリンター ON ならばプリンターにも入力文字を出す。

③ 受信ルーチン

- 行 290 入力ポートをしらべ、入力がなければ送信ルーチンにもどる。ホストから続々と送信してくると 350 行の PRINT 文で画面上に印字されていくが、その間にカーソルがチラチラすれば、余裕を持ってこのプログラムが走っていることを示すことになる。
- 行 300 入力から一字取り出す。
- 行 310 ACOS の TSS は行のはじめに“(7F)₁₆進”を数字送ってくるのでこれを読みとばす。
- 行 326 プリンター ON ならば、今読んだ文字をプリンターに送る。
- 行 330 “ 改行 ”文字を画面に出ないようにする。
- 行 340 受信が“ 復帰 ”ならばその前に“ 改行 ”をつけたす。
- 行 350 受信文字を画面に出す。ACOS の TSS は改行の時に、“ 復帰 ”、“ 改行 ”“(7F)₁₆進”、“(7F)₁₆進”……と送ってくる。この順序で画面に出すと、カーソルルーチンのために、前の行の初めの文字が消えてしまうので 330 行 340 行の操作を行なっている。これで問題は大体解消するか ACOS TSS がパスワードを画面から消す手続きをこのプログラムは余分な“ 改行 ”文字を入れるために正式に行わないことになる。
- 次の受信文字を受け取りに 290 行へ。

以上でプログラムのほねぐみの部分は終りである。以下割り込みルーチン、サブルーチンの説明に入る。

④ カーソルルーチン

行 370 カーソルの点滅中に割り込みが入ると、止まったカーソルが画面上に残ることがあるので、割り込みを保留する。

行 380 カーソルを出す位置を計算したのちAA%に入っているドット情報に従ってカーソルを画面に出す。PUT@はパソピアのBASICでドットを画面に出すための文で二度同じ場所に出すと丁度点滅する。

行 390 割り込みの保留を解除して、150行にもどる。

⑤ 一字消去ルーチン

行 410 送信時のミスタイプを修正するために“←”、“(DEL)”、“(CTR)H”の入力で、ホストに“(08)_{16進}”を送るようにする。これと同時にカーソルを一字分前に移動させる。カーソルを移動させても画面の文字は消えていないがホスト側では消されたものとして受け取っている。

⑥ 切断信号ルーチン

行 430 180行の判定で“(CTR)E”の入力でここにとんでくる。ここでは“(CTR)C”をホストに送っている。“(CTR)C”を直接入力するとBASICプログラムの中断となってしまうので、このように“(CTR)E”で代用する。

⑦ Breakルーチン

行 450 90行、100行で設定されているので[PF1]のKeyを押すとこの行にとんでくる。パソピアのT-BASICではポート番号(22)_{16進}に(04)_{16進}を送ると、ホストに向かってBreak信号が送られることになっている。ところが正常にBASICプログラムが走っている時には、このポートの内容はブロックされていてBASICプログラムでは変更できない。この理由で、このプログラムのBreakルーチンは少し複雑になっている。まずこの450行で行うエラー発生時の処理ルーチンを470行以下に、きりかえた後“!!break!!”の文字列を画面上に出しBreak key をプログラムが受けつけたことを示す。

行 460 80行で設定しておいたBreak文字を送り出す。ホストからの受信がなければこれでBreak信号がホストに送られるのでエラー処理をもとにもどしてもとの流れにもどる。

行 470 460行のBreak文字の送信時にホストからの受信があれば、I/Oエラーが発生するので、この行にとんでくる。そこで(22)_{16進}=(34)_{10進}のポートの内容を画面上に出して、プログラムがこの行に来たことを示す。不思議なことにとこの状態になると、ポート(22)_{16進}に信号(04)_{16進}が送れるようになる。

行 480 そこでFOR文を使ってタイミングを取りながら、ポート(22)_{16進}に(04)_{16進}を送

り出す。Break 信号の送り出しが終ると、一度ポートの中味が(04)₁₆進 になっていることをたしかめてRS232Cをもとの送受信が正常に行えるようにもどしておく。

行 490 画面のカーソル位置を行のはじめの位置にもどしておいてから460行の後半の文にもどる。

⑧ エラー処理ルーチン

行 510 110 行の設定で、通常のプログラムの走行時にエラーが発生すると、この行にとんで来る。カーソルルーチンの中でエラーが発生するとカーソルがとまったまま画面に残ることがおきるのでそれを消したのち

行 520 エラーの発生行、エラーの種類番号を画面に出しRS232Cを初期化して、もとの流れにもどる。

⑨ BASICのコマンドレベルにもどるルーチン

行 550 90 行、100 行の設定で **PF2** を押すとこの行にとんでくる。TSS 端末としての使用を中止するときに、この機能を使用する。この行でエラー処理の解除とファンクションKeyをもとにもどすことを行う。

行 560 この行でプリンターを普通の80字印字にもどしておく。

行 570 プログラムを終了させる。

⑩ ログオン文字をホストに送るルーチン

行 580 “\$\$\$CON, TSS”文字変数000\$に入れて、710行のライン出力のルーチンにとぶ。この文字を“\$\$\$CON, TSS ,, ASC”とすればASCIIモードのTSS端末として動作するようになる。

⑪ ハードコピールーチン

行 600 CALL BB%(=FFF7) 16進 でパソピアは画面のハードコピーをプリンターに送り出す。

⑫ プリンターON/OFFルーチン

行 620 **PF3** を押すとこの行にくる。LPP%=0はプリンターがOFF、LPP%=1が80字巾の印字、LPP%=2が136字巾の印字になるようにしてある。

行 660 **PF3** を押すたびに、この順にぐるぐるモードがかわる。これと同時に、カーソルのパターンを3種類に切り換えている。

⑬ ラインインプットルーチン

行 680 **PF4** Key を押すと200行からこの行にとんでくる。XP、YP に現在のカーソル位置を格納したのちLINE INPUT文を実行する。ここでカーソルのパターン

がBASIC のコマンドレベルのものにかわる。この状態ではパソコンの“ D EL ”, “ INS ” “ ← ”, “ → ”, “ ↑ ”, “ ↓ ” が使えるようになっているので、画面上の任意の行位置でホストに送りたい情報を作って、その行にカーソルを置いたまま“ 復帰 ”Keyを押す。ACOSのEDITOR SYSTEMでファイルの修正をしているときは、FV;n,FVS,P等のサブコマンドで修正したい行を画面に出しておいてRVサブコマンドを入れ“ ENTER ”と出たら \square でカーソルを上移動して、修正したい行の出ている位置に合わせて修正作業を行えばよい。又少し長い行を入力する時にもこの機能は便利である。710行のライン送信ルーチンにとび000\$の内容をホストに送り出す。

行 690 カーソルをもとの位置にもどし、もとの流れにもどる。

⑭ ライン送信ルーチン

行 710 ⑩、⑬、⑮の機能を使う時は、ホストに行単位で送信することが必要になる。T-BASICの文法上ではPRINT#-3,000\$とすればよいように見えるが000\$の内容が長いと(6字で50%、10字以上だとほぼ100%の確率で)I/Oエラーが発生する。この時ブレークルーチンのところでのべたポート(22)16進の内容が(04)16進にかわり、ホストに“ Break ”信号が送られてしまう。

行 720 このことをさけるために000\$を一字づつに分解して、

行 730 送り出す。

行 740 “ 復帰 ”を最後に送り出して、行の終りをホストにつけて、もとの流れにもどる。

⑮ フロッピー、ディスクからの転送ルーチン

行 760 オフラインで、フロッピーディスク上に作った、プログラム、データをホストに送り出す手続きは、つぎのようにする。

SYSTEM? FORT N“ 復帰 ”

* READ PPT,filename

READY \square PF4 押し下げ

とする。これでこの行のPRINT文が働く。

行 770,780 “ Input Y OR N”ときいてくるので上記の手続きをしてあれば“ Y ”か“ y ”を入力する。してなければNを入力すると送信ルーチンにもどるので上記の手続きをする。

行 790 “ Euter read filename ”ときいてくる。

行 800 これに対して“ FILES ”と答えるとドライブ2のファイル名を出して790行にもどる。

行810 790行の問い合せに“復帰”だけで答えると送信ルーチンにもどる。

行820 ディスク上のFile名が入力されると、そのFileをOpenして

行830 この行でFileの終りをカンシしながら

行840 L\$に1行づつ取り出す。

行850 その行を画面に出し、プリンターがONならばプリンターにも送り

行860 000\$に“改行”文字をつけたして、710行の行送信ルーチンで、ホストに送り出す。

行870 ファイルの終りを検出されると最後の行として“\$\$\$EOF”を送り出して、READ PPTの終りをホストにしらせる。

行880 ファイルをCLOSEして、送受信のループにもどる。ここで“*”がホストからくるのをまつ。フロッピーの中のFileの各行に、行番号がついていない時には、上記のPPTの後のfilename（ホスト側のファイル名）を必ずづけなければならない。これが省略されるとACOSのTSSではカレントファイルが受取りファイルになるが、この場合には行番号のない行は読みとばしてしまう。

⑩ フロッピーディスクにホストから受け取るルーチン

ホスト側のファイルからデータ又はプログラムをフロッピーに受け取るのには、つぎのようになる。

SYSTEM? FORT O filename…で取り出したいファイルをFORTRAN SYSTEMのカレントファイルにうつす。

* PF5 押し下げ

行900 この行でホストに“LIST”の文字を送り出すと、カレントファイルの中味が送られてくる。

行910 ポートに文字が入って来たら

行920 LINE INPUT文で行単位で受け取り、LF\$ (PP) にPPを1づつ増加しながら受け取る。

行930 FORTRAN STSTEMでのLISTコマンドでは最後の行のあとに空の行—実際には11字の画面にあらわれない文字つまり(7F) 16進や“改行”だけの行—を送ってくるのでこれでファイルの最後を検出する。

行940 “Enter write filename”ときいてくるのでフロッピー上につくるファイル名を入力する。

行970 ファイルをOPENして

行980,990 LF\$ (PP) の内容をファイルに書き込んでいく。

パソピアではフロッピーディスクへの書き込みの立ちあがりに時間を要する。これはT-BASICがディスクのマウントのための特別なコマンドを持たずに、必要に応じて自動的にこの作業するためであると思われる。この理由でホストのファイルをフロッピーにコピーを取るためには、ここでのべたようにバッファを置く必要があった。

4. おわりに

ここに紹介したプログラムは、多分に筆者の好みに合せたものになっている。プログラムは、フロッピーディスクに格納してあるので、御希望の方にはいつでもコピーを提供できる。御自分の好みに合せて、プログラムを作られるときの御参考になれば幸いである。

⑮、⑯のFDD \leftrightarrow TSS ファイルの機能を有効に使うためにはパソピアのツールとして TEXT EDITOR を作っておくと便利である。筆者はT-BASICで100行ほどのプログラムで“NEW”、“OLD”、“APPEND”、“FIND”、“COPY”、“CUT”、“PASTE”、“SAVE”P;n;L;n;B;n;F;nのように、ACOSのEDITOR に準じたコマンドを持ったプログラムを作製して利用している。

プログラムの作製にあたり、東芝パソコンサロンの大橋氏に、パソピアのマニュアルに書いていない使い方を種々教えていただいた。ここに記して感謝します。

表 1

パーソナルコンピューター	PASOPIA
グリーンディスプレイ	PA7150
ドットプリンター II	PA7251
ミニフロッピーディスクユニット	PA7200
音響カプラー	EPSON CP-20

```

10 'Terminal program for PASOPIA
20 'Initialization
30 DIM AA%(2),CC%(2),LF$(500)
40 CC%(0)=16:CC%(1)=1:CC%(2)=&H0
50 AA%(0)=16:AA%(1)=1:AA%(2)=&HFF
60 LP%=0:LPP%=0
70 BB%=&HFFF7
80 BRK$="$$$BRK"+CHR$(&HD)
90 ON KEY GOSUB 450,540,620,760,900
100 KEY(1) ON:KEY(2) ON:KEY(3) ON:KEY(4) ON:KEY(5) ON
110 ON ERROR GOTO 510
120 WIDTH 80:CLS:SCREEN 2:INIT -3,300,"E71"
130 'Send routine
140 K$=INKEY$
150 IF K$="" THEN GOSUB 370:GOTO 290
160 IF K$=CHR$(&H1D) THEN GOSUB 410:GOTO 140
170 IF K$=CHR$(&H8) THEN GOSUB 410:GOTO 140
180 IF K$=CHR$(&H5) THEN GOSUB 430:GOTO 290
190 IF K$=CHR$(&H1C) THEN GOTO 140
200 IF K$=CHR$(&H1E) THEN GOSUB 680:GOTO 290
210 IF K$=CHR$(&H1F) THEN GOTO 140
220 IF K$=CHR$(&H1) THEN GOSUB 580:GOTO 290
230 IF K$=CHR$(&H11) THEN GOSUB 600:GOTO 290
240 PRINT #-3,K$;:P$=INPUT$(1,#-3)
250 IF K$=CHR$(&HD) THEN K$=CHR$(&HA)+K$
260 PRINT K$;
270 IF LP% THEN LPRINT K$;
280 'Receive routine
290 IF PORT(-3)=0 THEN 140
300 R$=INPUT$(1,#-3)
310 IF R$=CHR$(&H7F) THEN 290
320 IF LP% THEN LPRINT R$;
330 IF R$=CHR$(&HA) THEN 290
340 IF R$=CHR$(&HD) THEN R$=CHR$(&HA)+R$
350 PRINT R$;:GOTO 290
360 'Cursor routine
370 KEY(0) STOP
380 X=POS(0)*8:Y=CSRLIN*8+7:IF X<623 THEN PUT@(X,Y),AA$:PUT@(X,Y),AA$
390 KEY(1) ON:KEY(2) ON:KEY(3) ON:KEY(4) ON:KEY(5) ON:RETURN
400 'Back space routine
410 PRINT #-3,CHR$(&H8);:P$=INPUT$(1,#-3):LOCATE POS(0)-1,CSRLIN:RETURN
420 'Disconnect routine
430 PRINT #-3,CHR$(&H3);:P$=INPUT$(1,#-3):RETURN
440 'Break routine
450 ON ERROR GOTO 470:PRINT " !!break!!";
460 PRINT #-3,BRK$;:P$=INPUT$(7,#-3):ON ERROR GOTO 510:RETURN 290
470 PRINT HEX$(INP(34));
480 FOR C=1 TO 100:OUT 34,&H4:NEXT C:PRINT HEX$(INP(34)):INIT -3,300,"E71"
490 PRINT CHR$(&HA);CHR$(&HD):RESUME NEXT
500 'Error routine
510 X=POS(0)*8:Y=CSRLIN*8+7:PUT@(X,Y),CC%,PSET
520 PRINT "ERL ";ERL;"ERR ";ERR:INIT -3,300,"E71":RESUME NEXT
530 'Back to BASIC
540 ON ERROR GOTO 0:KEY(0)OFF
550 IF LPP%=2 THEN LPRINT CHR$(&H1B);"N";:WIDTH LPRINT 80:
560 END
570 'Logon routine
580 OOO$="$$$CON,TSS":GOSUB 710:PRINT "$$$CON,TSS":RETURN
590 'Hard copy routine
600 X=POS(0)*8:Y=CSRLIN*8+7:PUT@(X,Y),CC%,PSET:CALL BB$:RETURN
610 'Printer on/off routine
620 IF LPP%=0 THEN LP%=-1:LPP%=1:AA%(2)=&HCC:RETURN

```

```

630 IF LPP%=2 THEN 650
640 LPP%=2:AA%(2)=&H81:LPRINT CHR$(&H1B);"Q";:WIDTH LPRINT 136:RETURN
650 LPP%=0:LP%=0:AA%(2)=&HFF
660 LPRINT CHR$(&H1B);"N";:WIDTH LPRINT 80:RETURN
670 'Edit mode routine
680 XP=POS(0):YP=CSRLIN:LINE INPUT OOO$:GOSUB 710
690 LOCATE XP,YP:RETURN
700 'Line output routine
710 FOR T=1 TO LEN(OOO$)
720 K$=MID$(OOO$,T,1)
730 PRINT #-3,K$;:P$=INPUT$(1,#-3):NEXT T
740 PRINT #-3,CHR$(&HD);:P$=INPUT$(1,#-3):RETURN
750 'Read file open routine
760 PRINT "Did you key in --FORT N-- and --READ PPT,file-name-- "
770 INPUT "Input Y OR N";YN$
780 IF YN$<>"y" AND YN$<>"Y" THEN 140
790 INPUT "Enter read file name";F$
800 IF F$="FILES" OR F$="files" THEN FILES 2:GOTO 790
810 IF F$="" THEN RETURN
820 OPEN F$ FOR INPUT AS #2
830 IF EOF(2) THEN 870
840 LINE INPUT #2,L$
850 PRINT L$:IF LP% THEN LPRINT L$
860 OOO$=L$+CHR$(&HA):GOSUB 710:GOTO 830
870 OOO$="$$$EOF":GOSUB 710
880 CLOSE #2:RETURN
890 'write fill open routine
900 PP=0:OOO$="LIST":GOSUB 710
910 IF PORT(-3)>0 THEN 920 ELSE 910
920 PP=PP+1:LINE INPUT#-3,LF$(PP):PRINT LF$(PP)
930 IF LEN(LF$(PP))=11 THEN 940 ELSE 910
940 INPUT "Enter write file name";F$
950 IF F$="FILES" OR F$="files" THEN FILES 2:GOTO 940
960 IF F$="" THEN RETURN
970 OPEN F$ FOR OUTPUT AS #2
980 FOR P=3 TO PP-1
990 PRINT #2,LF$(P):PRINT LF$(P):NEXT P:CLOSE #2:RETURN

```